

LUCRAREA NR. 1

STUDIUL SURSELOR DE CURENT

1. Scopul lucrării

În această lucrare se studiază prin simulare o serie de surse de curent utilizate în cadrul circuitelor integrate analogice: sursa de curent standard, sursa de curent simplă și sursa de curent Wilson.

2. Considerații teoretice cu privire la sursele de curent

În cadrul circuitelor integrate analogice sursele de curent sunt folosite ca elemente de polarizare (generează curenți de polarizare constanți) deoarece conduc la micșorarea sensibilității circuitului față de variațiile surselor de alimentare și ale temperaturii. Deoarece prezintă rezistențe de ieșire mari sursele de curent se folosesc în cadrul circuitelor integrate analogice și ca sarcini active în locul rezistențelor de valori mari, care necesită o arie de siliciu mai mare decât cea aferentă surselor de curent.

Cei mai importanți parametri ai unei surse de curent sunt: precizia raportului curenților prin diferite ramuri, rezistența de ieșire, stabilitatea curentului cu temperatura, independența curentului de tensiunea de alimentare.

Unele dintre sursele de curent cele mai utilizate în cadrul circuitelor integrate analogice sunt: sursa de curent standard, sursa de curent simplă și sursa de curent Wilson. Schemele acestor surse sunt prezentate în figura 1. În continuare se prezintă pe scurt unele caracteristici ale acestor surse de curent.

2.1. Sursa de curent standard

Schema sursei de curent standard este prezentată în figura 1(a). Este denumită în acest fel deoarece are schema identică cu cea folosită în circuitele cu componente discrete. Sursa de curent standard se utilizează pentru curenți mai mari decât 0,1 mA. Raportul dintre curentul de referință I_{ref} și cel de ieșire I_{out} este mai mare sau egal decât 5.

Tranzistorul T_1 este conectat ca diodă, având $U_{CB} = 0$ V. Astfel, tranzistorul T_1 se află în regiunea sa activă. El realizează compensarea termică a curentului I_{out} la variația tensiunii U_{BE} și a factorului β cu temperatura.




Fig. 1. Sursele de curent studiate: (a) sursa de curent standard, (b) sursa de curent simplă și (c) sursa de curent Wilson.

Presupunem că tranzistoarele T_1 și T_2 sunt identice și au β suficient de mare pentru a neglija curenții de bază față de curenții de colector. Curentul I_{ref} este egal cu:

$$I_{ref} = \frac{V_{cc} - U_{BE1}}{R_r + R_1}. \quad (1)$$

Raportul dintre curenții I_{out} și I_{ref} este aproximativ egal cu:

$$\frac{I_{out}}{I_{ref}} \cong \frac{R_1}{R_2}. \quad (2)$$

Dacă $R_1 = R_2$, atunci sursa este oglindă de curent.

Sursa de curent standard are dezavantajul dependenței curentului I_{out} de tensiunea de alimentare V_{cc} prin intermediul curentului I_{ref} . Conform relațiilor (1) și (2) rezultă că această dependență este practic liniară.

2.2. Sursa de curent simplă

Schema sursei de curent simple este prezentată în figura 1(b). Aceasta este un caz particular al sursei de curent standard în care $R_1 = R_2 = 0 \Omega$. Sursa de curent simplă este cea mai convenabilă pentru circuitele integrate analogice.

Din schema sursei de curent simple rezultă $U_{BE1} = U_{BE2}$. Prin urmare, avem:

$$U_T \ln\left(\frac{I_{c1}}{I_{s1}}\right) = U_T \ln\left(\frac{I_{out}}{I_{s2}}\right), \quad (3)$$

unde: I_{s1} și I_{s2} sunt curenții de saturație ai joncțiunilor bază-emitor ale tranzistoarelor T_1 și T_2 ; U_T este tensiunea termică, $U_T = kT/q$ ($\cong 26$ mV pentru $T = 300$ K).
Presupunem că tranzistoarele sunt identice și curenții de bază neglijabili. Astfel, rezultă:

$$\frac{I_{out}}{I_{ref}} \cong 1. \quad (4)$$

Deci, sursa este oglindă de curent.
Curentul I_{ref} este egal cu:

$$I_{ref} = \frac{V_{cc} - U_{BE1}}{R_r}. \quad (5)$$

În realitate curentul de colector, datorită efectului de modelare a grosimii bazei (efectul Early), este dat de relația:

$$I_c = I_s e^{\frac{U_{BE}}{U_T}} \left(1 + \frac{U_{CE}}{V_A}\right), \quad (6)$$

în care V_A este tensiunea Early. Ținând cont de relația de mai sus rezultă că:

$$\frac{I_{out}}{I_{ref}} \cong \frac{1 + \frac{U_{CE2}}{V_A}}{1 + \frac{U_{BE1}}{V_A}}. \quad (7)$$

Datorită efectului Early factorii β ai celor două tranzistoare nu sunt egali. Astfel, dacă două tranzistoare identice lucrează la tensiuni U_{CE} diferite (de exemplu $U_{CE1} < U_{CE2}$), atunci și factorii β vor fi diferiți ($\beta_1 < \beta_2$). În cazul sursei de curent simple $U_{CE1} = U_{BE1} \cong 0,6$ V $< U_{CE2}$. Prin urmare, rezultă $\beta_1 < \beta_2$. Raportul I_{out}/I_{ref} este aproximativ egal cu:

$$\frac{I_{out}}{I_{ref}} \cong \frac{\beta_2}{\beta_1 + 2}. \quad (8)$$

De regulă, raportul I_{out}/I_{ref} are valori cuprinse între 1,1 și 1,2.

Sursa de curent simplă, la fel ca și sursa de curent standard, are dezavantajul dependenței curentului I_{out} de tensiunea de alimentare V_{cc} prin intermediul curentului I_{ref} , care pe baza relațiilor (5) și (7) este practic liniară.

2.3. Sursa de curent Wilson

Schema sursei de curent simple este prezentată în figura 1(c). Această sursă menține constant curentul I_{out} datorită reacției negative serie-paralel realizată prin tranzistorul T_2 (este o reacție negativă datorită inversării de fază produse de T_2 (în conexiune EC)).

Curentul I_{ref} este egal cu:

$$I_{ref} = \frac{V_{cc} - U_{BE1} - U_{BE2}}{R_r}. \quad (9)$$

Tensiunea $U_{CE3} \cong 0,6$ V, iar tensiunea $U_{CE2} = U_{BE1} + U_{BE2} \cong 1, 2$ V. Între tensiunile U_{CE} ale tranzistoarelor există inegalitatea $U_{CE1} > U_{CE2} > U_{CE3}$. Prin urmare și aici intervine efectul Early, rezultând $\beta_1 > \beta_2 > \beta_3$, cu observația că β_2 și β_3 sunt foarte apropiate între ele deoarece tensiunile U_{CE2} și U_{CE3} sunt apropiate între ele.

Presupunem că toate tranzistoarele sunt identice. Raportul I_{out}/I_{ref} este aproximativ egal cu:

$$\frac{I_{out}}{I_{ref}} \cong \frac{\beta_1\beta_3 + 2\beta_1}{\beta_1\beta_2 + \beta_2 + \beta_3 + 2} \cong \frac{\beta_1\beta_3}{\beta_1\beta_2} = \frac{\beta_3}{\beta_2}. \quad (10)$$

Se obține un raport I_{out}/I_{ref} mai mic decât 1, dar foarte apropiat de 1. Prin urmare, sursa este oglindă de curent.

Datorită avantajelor pe care le prezintă (curentul I_{out} este menținut constant și un raport I_{out}/I_{ref} foarte apropiat de 1) sursa de curent Wilson este mult mai performantă decât sursa de curent simplă.

Și această sursă are dezavantajul dependenței curentului I_{out} de tensiunea de alimentare V_{cc} prin intermediul curentului I_{ref} . Pe baza relațiilor (9) și (10) rezultă că această dependență este practic liniară.

3. Desfășurarea lucrării

Pentru ca simularea să fie cât mai apropiată de realitate se consideră că tranzistoarele sunt din cadrul circuitului integrat analogic CA 3086 – arie de tranzistoare npn.

Schemele folosite în cadrul analizelor SPICE sunt prezentate în figura 2 (sunt prezentate, de asemenea, și nodurile aferente).




Fig. 2. Schemele folosite în cadrul analizelor SPICE pentru: (a) sursa de curent standard, (b) sursa de curent simplă și (c) sursa de curent Wilson.

3.1. Sursa de curent standard

Schema folosită în cadrul analizelor SPICE este prezentată în figura 2(a). Programul SPICE utilizat pentru studiul sursei de curent standard (fișierul *ex1.cir* din subdirectorul...\\Work\\Lucrari\\Lucrarea1) este următorul:

Sursa de curent standard

```
Vcc 1 0 DC 15 ; Vcc = 15 V
Rr 1 2 2.6kohm ; Rr = 2.6 kΩ
Rs 1 4 5kohm ; Rs = 5 kΩ
Q1 2 2 3 CA3086 ; Tranzistoarele Q1 și Q2 sunt din cadrul circuitului integrat CA 3086
Q2 4 2 5 CA3086
R1 3 0 1kohm ; R1 = 1 kΩ
R2 5 0 5kohm ; R2 = 5 kΩ
```

```
.MODEL CA3086 NPN ; modelul tranzistoarelor din cadrul circuitului integrat CA 3086
+ IS=10.000E-15 BF=156.66 VAF=100 IKF=36.783E-3
+ ISE=114.82E-15 NE=1.4743 BR=.1001 VAR=100
+ IKR=10.010E-3 ISC=10.000E-15 RC=10 CJE=1.0260E-12
+ MJE=.33333 CJC=991.79E-15 MJC=.33333 TF=278.55E-12
+ XTF=91.905 VTF=18.899 ITF=.77631 TR=10.000E-9
```

* Analize SPICE

```
.PROBE I(Rs) ; se vizualizează curentul Iout (I(Rs))
.DC Vcc 5 25 1 ; tensiunea Vcc variază între valorile 5 V și 25 V cu pasul de 1 V
.END
```

Se efectuează următoarele:

a) Pentru programul dat se reprezintă grafic curentul de ieșire I_{out} ($I(R_s)$) în funcție de tensiunea de alimentare V_{cc} ($V(1)$).

b) Se realizează o analiză parametrică .STEP, în care parametrul este valoarea lui R_2 . Aceasta ia, pe rând, valorile întregi din cadrul intervalului [5, 14] k Ω . În acest scop, linia de descriere a rezistenței R_2 se înlocuiește cu:

```
R2 5 0 {val}.
```

De asemenea, după modelul tranzistoarelor, în locul analizelor SPICE existente, se scrie:

```
.Param val = 5k          ; val = 5 k
* Analize SPICE
.DC Vcc 15 15 1          ; Vcc = 15 V
.STEP Param val list 5k 6k 7k 8k 9k 10k 11k 12k 13k 14k      ; analiza parametrică în care
                        ; parametrul val ia, pe rând, valorile specificate în cadrul listei
.PRINT DC I(Rs) I(Rr)    ; se tipăresc în fișierul de ieșire ex1.out, pentru fiecare valoare a lui
                        ; R2, valorile curenților I(Rs) și I(Rr)
.END
```

Se reprezintă grafic folosind programul MATLAB raportul I_{ref}/I_{out} în funcție de valoarea lui R_2 , atât folosind datele obținute prin simulare, cât și cele obținute pe baza relației (2). Se compară, prin intermediul erorii relative rezultatele obținute pe baza simulării cu cele obținute prin aplicarea relației (2) (valoarea adevărată se consideră cea obținută pe baza simulării). Ce concluzie se poate deduce din această comparație?

3.2. Sursa de curent simplă

Schema folosită în cadrul analizelor SPICE este prezentată în figura 2(b). Programul SPICE utilizat pentru studiul sursei de curent standard (fișierul *ex2.cir* din subdirectorul...\\Work\\Lucrari\\Lucrarea1) este următorul:

Sursa de curent simpla

```
Vcc 1 0 DC 15          ; Vcc = 15 V
Rr 1 2 14.4kohm        ; Rr = 14.4 k $\Omega$ 
Rs 1 3 8kohm           ; Rs = 8 k $\Omega$ 
Q1 2 2 0 CA3086        ; Tranzistoarele Q1 și Q2 sunt din cadrul circuitului integrat CA 3086
Q2 3 2 0 CA3086
```

```
.MODEL CA3086 NPN      ; modelul tranzistoarelor din cadrul circuitului integrat CA 3086
+ IS=10.000E-15 BF=156.66 VAF=100 IKF=36.783E-3
+ ISE=114.82E-15 NE=1.4743 BR=.1001 VAR=100
+ IKR=10.010E-3 ISC=10.000E-15 RC=10 CJE=1.0260E-12
+ MJE=.33333 CJC=991.79E-15 MJC=.33333 TF=278.55E-12
+ XTF=91.905 VTF=18.899 ITF=.77631 TR=10.000E-9
```

* Analize SPICE

```
.PROBE I(Rs)          ; se vizualizează curentul Iout (I(Rs))
.DC Vcc 5 25 1        ; tensiunea Vcc variază între valorile 5 V și 25 V cu pasul de 1 V
.END
```

Se efectuează următoarele:

a) Pentru programul dat se reprezintă grafic curentul de ieșire I_{out} ($I(R_s)$) în funcție de tensiunea de alimentare V_{cc} ($V(1)$).

b) Se realizează o analiză parametrică .STEP, în care parametrul este valoarea lui β (BF în modelul tranzistorului). Aceasta ia, pe rând, valorile: 25, 50, 75, 100, 125, 150, 175 și 200. În acest scop în cadrul modelului tranzistoarelor se introduce $BF = \{val\}$. De asemenea, după modelul tranzistoarelor, în locul analizelor SPICE existente, se scrie:

```
.Param val = 156.66 ; val = 156,66
* Analize SPICE
.DC Vcc 15 15 1 ; Vcc = 15 V
.STEP Param val list 25 50 75 100 125 150 175 200 ; analiza parametrică în care
; parametrul val ia, pe rând, valorile specificate în cadrul listei
.PRINT DC I(Rs) I(Rr) ; se tipăresc în fișierul de ieșire ex1.out, pentru fiecare valoare a lui
; R2, valorile curenților I(Rs) și I(Rr)
.END
```

Se reprezintă grafic folosind programul MATLAB raportul I_{out}/I_{ref} în funcție de valoarea lui β pe baza datele obținute prin simulare. Să se explice rezultatele obținute.

c) Se revine la $BF = 156,66$ în cadrul modelului tranzistorului. Se realizează o analiză parametrică .STEP, în care parametrul este valoarea tensiunii Early (VAF în modelul tranzistorului). Aceasta ia, pe rând, valorile: 25 V, 50 V, 75 V, 100 V, 125 V, 150 V, 175 V și 200 V. În acest scop în cadrul modelului tranzistoarelor se introduce $VAF = \{val\}$. De asemenea, după modelul tranzistoarelor, în locul analizelor SPICE existente, se scrie:

```
.Param val = 100 ; val = 100
* Analize SPICE
.DC Vcc 15 15 1 ; Vcc = 15 V
.STEP Param val list 25 50 75 100 125 150 175 200 ; analiza parametrică în care
; parametrul val ia, pe rând, valorile specificate în cadrul listei
.PRINT DC I(Rs) I(Rr) VC(Q1) VC(Q2) ; se tipăresc în fișierul de ieșire ex1.out, pentru fiecare
; valoare a lui R2, valorile curenților I(Rs) și I(Rr), precum și
; ale tensiunilor UCE1 (= UBE1) și UCE2.
.END
```

Se reprezintă grafic folosind programul MATLAB raportul I_{out}/I_{ref} în funcție de valoarea tensiunii V_A , atât pe baza valorilor I_{out} și I_{ref} obținute prin simulare, cât și pe baza valorilor calculate cu relația (7), în care U_{BE1} și U_{CE2} se citesc din fișierul *ex2.out*. Se compară, prin intermediul erorii relative rezultatele obținute pe baza celor două proceduri (valoarea adevărată se consideră cea obținută folosind I_{out} și I_{ref} din fișierul *ex2.out*). Ce concluzie se poate deduce din această comparație?

3.3. Sursa de curent Wilson

Schema folosită în cadrul analizelor SPICE este prezentată în figura 2(c). Programul SPICE utilizat pentru studiul sursei de curent standard (fișierul *ex3.cir* din subdirectorul...\\Work\\Lucrari\\Lucrarea1) este următorul:

Sursa de curent Wilson

Vcc 1 0 DC 15 ; $V_{cc} = 15 \text{ V}$
Rr 1 2 13.8kohm ; $R_r = 13.8 \text{ k}\Omega$
Rs 1 4 8kohm ; $R_s = 8 \text{ k}\Omega$
Q1 4 2 3 CA3086 ; Tranzistoarele $Q_1 - Q_3$ sunt din cadrul circuitului integrat CA 3086
Q2 2 3 0 CA3086
Q3 3 3 0 CA3086

.MODEL CA3086 NPN ; modelul tranzistoarelor din cadrul circuitului integrat CA 3086
+ IS=10.000E-15 BF=156.66 VAF=100 IKF=36.783E-3
+ ISE=114.82E-15 NE=1.4743 BR=.1001 VAR=100
+ IKR=10.010E-3 ISC=10.000E-15 RC=10 CJE=1.0260E-12
+ MJE=.33333 CJC=991.79E-15 MJC=.33333 TF=278.55E-12
+ XTF=91.905 VTF=18.899 ITF=.77631 TR=10.000E-9

* Analize SPICE

.PROBE I(Rs) ; se vizualizează curentul I_{out} ($I(R_s)$)
.DC Vcc 5 25 1 ; tensiunea V_{cc} variază între valorile 5 V și 25 V cu pasul de 1 V
.END

Se efectuează următoarele:

- Pentru programul dat se reprezintă grafic curentul de ieșire I_{out} ($I(R_s)$) în funcție de tensiunea de alimentare V_{cc} ($V(1)$).
- Se reprinted grafic raportul I_{out}/I_{ref} în funcție de valoarea lui β . În acest scop se procedează la fel ca și la punctul 3.2(b). Se compară rezultatele obținute pentru sursa de curent Wilson și pentru sursa de curent simplă. În acest scop rezultatele obținute pentru cele două surse de curent se reprezintă pe același grafic.

Întrebări:

- Unde sunt utilizate sursele de curent în cadrul circuitelor integrate analogice ?
- Să se deducă relațiile (2) și (10).
- De ce sursa de curent Wilson este mai performantă decât sursa de curent simplă ?